TEMPLATE = app
CONFIG += console c++11
CONFIG -= app_bundle
CONFIG -= qt
INCLUDEPATH += ../../../tb_interface
DESTDIR = $$OUT_PWD/../../../bin/modules

include (../../../taskbusplatform.pri)

SOURCES += \
	main.cpp \
    function_info.cpp

DISTFILES += \
    transform_fft.json

win32{
    INCLUDEPATH +="$$PWD/../../3rdlibs/win32/fftw"
    contains(QT_ARCH, i386) {
	LIBS+=-L"$$PWD/../../3rdlibs/win32/fftw/x86" -llibfftw3-3
    # 第三方库部署配置 - FFTW动态库
    QMAKE_PRE_LINK += $${QMAKE_COPY} "$$PWD/../../3rdlibs/win32/fftw/x86/*fftw*.dll" $$DESTDIR
    } else {
    LIBS+=-L"$$PWD/../../3rdlibs/win32/fftw/x64" -llibfftw3-3
    # 第三方库部署配置 - FFTW动态库
    QMAKE_PRE_LINK += $${QMAKE_COPY} "$$PWD/../../3rdlibs/win32/fftw/x64/*fftw*.dll" $$DESTDIR
    }
}else: LIBS+=-lfftw3

QMAKE_POST_LINK += $${QMAKE_COPY} $$PWD/readme.md $$DESTDIR/$${TARGET}.md
